
보통 스크립트를 사용한 무비에서는 프레임 번호부터 사용자의 암호, 우주선에

서발사한포탄어뢰의속도에이르기까지모든것을직접추적하고조작해야한다.

그런정보를모두관리하기위해그값들을액션스크립트의기본정보저장소인‘변

수(variable)’에저장해야한다.

변수는 돈 대신 정보(데이터)가 들어 있는 은행 통장이라고 생각할 수 있다. 새

로운 변수를 만드는 것은 새로운 계좌를 만드는 것과 같다. 즉 나중에 필요한 것을

저장해둘장소를만드는것이다. 그리고모든은행계좌에계좌번호가있는것처럼

모든변수에는그변수와연관된이름이있으며, 그 이름을이용하여변수에저장된

데이터를액세스할수있다.

변수를 만들고 나면 우리가 원하는 대로 새로운 데이터를 집어넣을 수 있다. 은

행계좌에돈을입금하는것과비슷하다고보면된다. 또는변수의이름을이용하여

변수에 어떤 값이 들어 있는지 알아낼 수 있다. 마치 은행 계좌의 잔고를 확인하는

것과비슷하다. 변수가더이상필요없다면변수를지움으로써은행계좌를없애듯

이변수를없앨수도있다.

이때가장주목할만한기능은변수를이용하여무비가재생되는동안변하거나

계속 새로 계산되는 데이터를 참조하는 기능이다. 은행 계좌에 있는 잔액이 변해도

계좌번호는그대로남아있는것처럼그안에있는데이터가바뀌어도변수이름은

바뀌지 않는다. 변화하는 내용물에 대해 고정된 이름을 사용하면 복잡한 계산을 할

수도 있고 카드 게임에서 카드를 추적할 수도 있으며, 방명록을 저장할 수도 있고

변화하는조건에따라플레이헤드를다른위치로옮길수도있다.

기대감으로 가득 차서 눈빛이 번뜩이는 독자들도 있을 것이다. 사실 은행에 관

한 비유를 읽으면서 독자들이 책을 덮지나 않았을까 걱정된다. 이제 변수를 만드는

것부터시작해서변수에대한모든것을알아보기로하자.

변수 만들기(선언)
변수를 새로 만드는 것을‘선언(declaration)’이라고 한다. 선언은 은행에 비유

하자면 구좌 개설에 해당한다. 변수를 선언하는 순간부터 변수는 그 프로그램 내에

존재하게된다. 변수를처음선언했을때는그변수는비어있다. 이렇게비어있는

변수에 우리가 새로운 값을 집어넣게 된다. 이 상태에서 변수에는 undefined(데이

터가들어있지않음을의미함)라는특별한값이저장되어있다.

새로운변수를선언할때는 var 선언문을사용한다. 예를들면다음과같다.

var speed;

var bookTitle;

var x;

var라는 단어는 인터프리터에 변수를 새로 선언하는 중이며 그 뒤에 오는

speed, bookTitle, x와같은텍스트가새로운변수의이름이라는것을알려준다. 변

수는 키프레임, 버튼, 무비 클립과 같이 코드를 만들 수 있는 곳이라면 어디서든지

선언할수있다.

다음과같이하나의 var 선언문에서여러개의변수를동시에선언할수도있다.

var x, y, z;

하지만이렇게하면각변수옆에주석을추가하기힘들어진다.

변수를 만들고 나면 그 변수에 값을 대입할 수 있지만 변수 대입을 배우기 전에

변수선언을좀더자세히살펴보자.

자동 변수 생성

데이터를변수에저장하기전에반드시변수를먼저선언해야하는언어가많이

있다. 그러한 언어에서는 변수를 선언하지 않고 사용하면 오류가 발생한다. 하지만

액션스크립트는 그다지 까다롭지 않다. 존재하지 않는 변수에 어떤 값을 대입하면

인터프리터에서자동으로새로운변수를만들어준다. 은행에계속비유하자면처음

으로돈을입금할때자동으로새로운계좌를만들어주는것과같다.

이러한 방법은 편리하긴 하지만 단점도 있다. 변수를 직접 선언하지 않으면 코

드를점검할때어떤기준이될만한것이없다. 게다가 var 선언문을이용하여직접

변수를선언하면자동으로변수가선언되는것과는다른결과가나올수도있다. 이

책에서는먼저변수를선언한뒤에사용하는데, 이렇게하는것이더안전하다.

사용할 수 있는 변수 이름

변수를 만들기 전에 변수 이름은 다음과 같은 조건을 만족시켜야 한다는 점에

주의하자.

䤎 영문자, 숫자, 밑줄만을 이용한다(공백, 하이픈, 문장 부호 따위는 사용하면

안된다).

䤎 문자또는밑줄로시작해야한다.

䤎 255자를 넘기면 안 된다(물론 255자를 넘기기는 쉽지 않지만 그렇게 될 것 같

다면변수이름붙이는방법을바꾸는게좋을것이다).

䤎 대소문자를 구분한다(실제로 대소문자를 구분하는 것은 아니지만 프로그래밍

할때대소문자를일관되게사용하는것이좋다).

다음과같은변수이름은올바른변수이름이다.

var first_name;

var counter;

var reallyLongVariableName;

하지만다음과같은변수이름은옳지않으며, 이렇게하면오류가발생한다.

var 1first_name;                 // 숫자로 시작
var variable name with spaces; // 공백이 들어 있음
var another-illegal-name; // 하이픈이 들어 있음

거의 쓰이지 않지만 프로그래밍을 통해 동적으로 변수 이름을 정할 수도 있다.

표현식으로부터변수이름을만들려면 set 선언문을이용하면된다. 예를들어아래

코드에서는 player1name이라는변수에 bruce라는값을대입한다.

var i = 1;

set ("player" + i + "name", "bruce");

나중에 배울 배열이나 객체를 이용하면 동적으로 이름이 주어지는 데이터를 더

편하게관리할수있으므로, 동적으로이름이주어지는변수이름대신배열이나객

체를이용하는것이좋다.

처음에 변수 선언하기

무비프리로더뒤에오는첫째키프레임인무비의주요스크립트시작부분에서

변수를 선언하는 습관을 들이도록 하자. 나중에 알아보기 좋도록 각 변수의 용도를

주석으로 달아 놓는 것도 잊지 말자. 잘 짜여진 스크립트라면 보통 다음과 같은 식

으로시작된다.

// ^^^^^^^^^^^^^^^^^^^^

// 변수 선언
// ^^^^^^^^^^^^^^^^^^^^

var ballSpeed;   // 공의 속도, 최고 10

var score;       // 현재 점수

var hiScore;     // 최고 점수(세션간에 저장되지 않는다).

var player1;     // palyer1의 이름, 사용자가 입력함

변수를만들면서동시에초기값을설정할수도있다.

var ballSpeed = 5;   // 공의 속도를 5로 설정
var score = 0;       // 현재 점수
var hiScore = 0;     // 최고 점수(세션간에 저장되지 않는다).

변수 값 대입
이제 조금 더 흥미로운 내용인 변수에 데이터를 대입하는 방법을 배워보자. 여

기서도은행에비유하자면계좌에돈을입금하는것과같다고말할수있다. 변수에

값을대입하려면다음과같이하면된다.

variableName = value;

여기서 variablename은 변수 이름이고 value는 변수에 저장되는 데이터이다.

이러한구문을사용하는방법은다음과같다.

bookTitle = "ActionScript: The Definitive Guide";

등호의 왼쪽에 있는 bookTitle은 변수의 이름(인식자)이다. 등호 오른쪽에는

ActionScript: The Definitive Guide라는 구문이 있는데, 이것이 변수의 값(저장하

고자 하는 데이터)이 된다. 등호는 대입 연산자라고 부른다. 이 연산자는 등호 오른

쪽에 있는 것을 왼쪽에 있는 변수에 대입하겠다는 것을 인터프리터에 알려준다. 만

약 왼쪽에 있는변수가아직만들어지지않은변수라면플래시에서알아서그변수

를선언한다(하지만이러한방법은그다지바람직하지못하다).

변수에값을대입하는예를두가지더들어보자.

speed = 25;

output = "thank you";

첫째 예에서는 speed라는 변수에 25라는 정수를 대입한다. 변수에는 텍스트뿐

만아니라숫자도저장할수있다. 잠시후에또다른종류의데이터도대입할수있

다는것을배울것이다. 둘째예에서는 thank you라는텍스트를 output이라는변수

에대입한다. 액션스크립트에서는텍스트문자열을나타낼때큰따옴표(“”)를 사용

한다.

이제 y에 1 + 5라는표현식값을대입하는조금더복잡한예를들어보자.

y = 1 + 5;

y = 1 + 5;라는 선언문이 실행되면 우선 1을 5에 더하여 6을 계산한 다음 y에

6을 대입한다. 오른쪽에 있는 표현식은 왼쪽에 있는 변수에 값을 대입하기 전에 계

산된다. 아래예에서는 y가포함된표현식을다시다른변수인 z에대입한다.

z = y + 4;

여기서도등호오른쪽에있는표현식을먼저계산하고그값을 z에대입한다. 인

터프리터에서는 y의현재값을(말하자면계좌의잔고를) 구해서 거기에 4를더한다.

y 값은 6이기때문에 z에는 10을대입한다.

데이터형에상관없이어떤데이터숫자, 텍스트및다른데이터형을사용해도변

수에데이터를대입하는방법은거의똑같다. 예를들면아직배열을배우진않았지

만다음과같은문장이변수를대입하는선언문이라는것은쉽게알수있을것이다.

myList = ["John", "Joyce", "Sharon", "Rick", "Megan"];

앞에서와 마찬가지로 변수 이름이 왼쪽으로 가고 대입 연산자(등호)가 가운데

있고대입하려는값이오른쪽에있다.

같은 값을 여러 개의 변수에 한꺼번에 대입하고 싶다면 다음과 같이 연속으로

대입할수도있다.

x = y = z = 10;

변수 대입은 언제나 오른쪽에서 왼쪽으로 진행된다. 위 선언문에서는 10이 z에

대입되고 z 값이 y에대입되고, y 값이 x에대입된다.

변수 값 변경 및 검색
변수를만든후에는 [예제 2-1]에서 볼 수 있듯이새로운값을계속해서대입할

수있다.

[예제 2-1] 변수 값 바꾸기

var firstName;                   // firstName을 선언한다.

firstName = "Graham";            // firstName 값을 설정한다.

firstName = "Gillian";           // firstName 값을 바꾼다.

firstName = "Jessica";           // firstName 값을 다시 바꾼다.

firstName = "James";             // firstName 값을 또 바꾼다.

var x = 10;                      // x를 선언하고 숫자를 대입한다.

x = "loading...please wait...";  // x에 텍스트를 대입한다.

x의 데이터형을 숫자에서 텍스트 데이터로 바꿀 때도 단지 원하는 형의 데이터

를대입하기만하면된다는점을기억해두자. 프로그래밍언어에따라변수의데이

터형을바꿀수없는것도있지만, 액션스크립트에서는마음대로바꿀수있다.

변수를 만들고 그 변수에 값을 대입하는 것은 변수 값을 나중에 다시 알아낼 수

없다면무의미한일이다. 변수에저장된값을구하려면그냥변수값이필요한곳에

서 변수 이름을 쓰기만 하면 된다. 변수 이름이 나타날 때마다(변수를 선언할 때와

대입선언문의왼쪽에있는경우는제외) 그 이름이변수값으로변환된다. 아래예를

살펴보자.

newX = oldX + 5;  // newX에 oldX와 5를 더한 값을 대입
ball._x = newX;   // ball 무비 클립의 수평 위치를 newX 값으로 설정
trace(firstName); // firstName의 값을 Output 창에 출력

ball._x에서 ball은 무비 클립 이름이고 ._x는 그 클립의 x축 속성(스테이지에서

의수평위치)을의미한다. 속성에관한것은나중에자세히배우게될것이다. 마지

막 줄에 있는 trace(firstName)에서는 스크립트가 실행되는 도중에 변수 값을 화면

에표시해주는데, 코드를디버깅할때편리하게쓰인다.

변수에 값이 있는지 확인하기

종종변수에있는값을알아보기전에변수에어떤값이들어있는지확인하고싶

을 때가 있다. 앞에서 배운 것처럼 선언은 했지만 아직 아무런 값도 대입하지 않은

변수에는 undefined라는 특별한 값이 들어 있다. 변수에 어떤 값을 대입한 적이 있

는지확인하려면아래와같이그변수값을 undefined라는키워드와비교하면된다.

if (someVariable != undefined) {

// someVariable에 어떤 값이 들어있을 때에만 여기에 있는 코드를 실행한다.

}

부등연산자(!=)는두값이서로다른지확인할때쓰인다.

값 유형
액션스크립트프로그래밍에서쓰이는데이터에는다양한형이있다. 지금까지는

숫자와 텍스트만 사용했지만 데이터형에는 부울형, 배열, 함수, 객체 같이 다양한

종류가있다. 각 데이터형을자세히살펴보기전에우선변수사용과관련된문제에

대해짚고넘어가기로하자.

자동 유형

모든액션스크립트변수에는임의의데이터형을가진데이터를저장할수있다.

어떻게 보면별것아니라고생각할수도있겠지만모든변수에임의의데이터형을

가진 데이터를 저장할 수 있는 언어는 생각보다 흔하지 않다. C++나 자바 같은 언

어에서는유형이정해진변수를사용하기때문에, 모든변수에변수를선언할때지

정한한가지유형의데이터만저장할수있다. 액션스크립트변수는데이터를변수

에대입할때인터프리터에서변수의데이터형을직접설정하는방식으로자동적으

로형이결정된다.

액션스크립트에서는변수에임의의데이터형을저장할수있을뿐만아니라, 변

수의 데이터형을 동적으로 바꿀 수도 있다. 변수에 원래 저장되어 있던 값과 다른

데이터형을 가진 새로운 값을 대입하면 변수의 형이 자동으로 바뀐다. 따라서 액션

스크립트에서는아래와같은스크립트를사용해도문제가없다.

x = 1;                   // 숫자
x = "Michael";           // 문자열
x = [4, 6, "hello"];     // 배열
x = 2;                   // 다시 숫자

C++나자바같은언어에서는자동으로새로운데이터형으로변경하는것이불가

능하므로 데이터가 변수의 원래 데이터형으로 변환된다(변환이 불가능한 경우에는

오류가발생한다). 액션스크립트의변수는자동유형및동적유형기능때문에앞으

로나오는절에서설명할몇가지중요한특징을지닌다.

값 자동 변환

상황에따라액션스크립트에서특정한형의데이터가필요한경우가있다. 만약

필요한형과맞지않는값을가진변수를사용하면인터프리터에서그데이터를변

환한다. 예를 들어 숫자가 필요한 위치에 텍스트 변수를 사용하면 인터프리터에서

는필요에따라그변수의텍스트값을숫자값으로변환한다. [예제 2-2]를 보면 z

값이 2로 설정된다. 이렇게 되는 이유는 뺄셈 연산에서는 숫자가 필요하기 때문에,

y 값이 4라는 문자열에서 숫자 4로 변환되어 6(x 값)에서 4를 뺀 2가 z에 저장되기

때문이다.

[예제 2-2] 문자열에서 숫자로 자동 변환

x = 6;      // x는 숫자 6

y = "4";    // y는 문자열 "4"

z = x - y;  // 이렇게 하면 z 값이 2가 된다.

반대로 숫자 변수를 문자열이 들어갈 자리에 사용하면 인터프리터에서 숫자를

문자열로변환한다. [예제 2-3]에서 z는숫자 10이아니라문자열“64”가된다. x +

y에있는두번째피연산자가문자열이기때문이다. 따라서 (+) 연산자는덧셈이아

닌문자열을합치는연산을처리하게된다. x 값(6)은 문자열 6으로변환되고그문

자열이 4(y의값)와합쳐져 64라는결과가나온다.

[예제 2-3] 숫자에서 문자열로 자동 변환

x = 6;      // x 는 숫자 6

y = "4";    // y 는 문자열 "4"

z = x + y;  // 이렇게 하면 z 값이 64가 된다.

어떤 표현식에서 변수를 사용할 때는 자동 형 변환이 변수에 있는 데이터의 복

사본에대해이루어지기때문에, 원래변수의데이터형에는영향을주지않는다. 변

수형은변수에원래데이터형과다른유형의데이터를대입할때만바뀐다. 따라서

[예제 2-2]와 [예제 2-3]을 실행하더라도 y는그대로문자열로, x는그대로숫자로

남는다.

셋째줄에있는연산자([예제 2-2]에서는 -, [예제 2-3]에서는+) 때문에, z에대

입되는 값이 크게 달라진다는 점에 주의하자. [예제 2-2]에서는 4라는 문자열이 숫

자 4로바뀌지만 [예제 2-3]에서는반대로숫자 6이문자열 6으로바뀐다. 데이터형

변환규칙이+ 연산자와 - 연산자에따라서로다르기때문이다. 데이터변환규칙

에 대해서는‘3장. 데이터와 데이터형’에서, 연산자에 대해서는‘5장. 연산자’에서

자세히다룰것이다.

수동으로 형 검사하기

자동 유형과 값 자동 변환이 편리하긴 하지만 [예제 2-2]나 [예제 2-3]에서 볼

수 있듯이 예상하지 못한 결과가 나올 수도 있다. 여러 가지 데이터형이 섞인 경우

에는명령을수행하기전에 typeof 연산자를이용하여변수의데이터형을알아보는

것이좋을수도있다.

productName = "Macromedia Flash";  // 문자열 값
trace(typeof productName);         // "string"이 출력된다.

변수 형을 알고 나면 상황에 따라 적절한 조치를 취할 수 있다. 예를 들어 아래

코드에서는어떤변수가숫자인지미리확인하고있다.

if (typeof age == "number"){

// 계속 해도 괜찮다.

} else {

trace ("Age isn't a number");  // 오류 메시지 출력
}

typeof 연산자에대한자세한내용은 5장을참조하기바란다.

변수 영역
지금까지 플래시 문서의 메인 타임라인에 있는 하나의 프레임에서 변수를 만들

고 그 값을 검색하는 것에 대해 살펴보았다. 하나의 문서에 여러 개의 프레임과 여

러개의무비클립타임라인이들어있으면변수를만들고검색하는것이좀더복잡

해진다.

그이유를알아보기위해몇가지시나리오를생각해보자.

시나리오 1

x라는변수를메인타임라인의 1번프레임에서만드는경우를가정해보자. x를

만든후에 10을대입해보자.

var x;

x = 10;

그리고나서 2번프레임에아래코드를집어넣는다.

trace(x);

그리고나서무비를실행시켜보자. Output 창에무엇이나타나게될까? 변수를

만든위치는 1번프레임이지만그값을검색하는위치는 2번프레임이다. 이런경우

에도변수가그대로있을까? 답은‘그렇다’이다.

시나리오 2

시나리오 1과 마찬가지로 x를 만들고 그 값을 설정했다고 가정해 보자. 하지만

이번에는변수를설정하는코드를 1번프레임에직접넣지않고 1번프레임에있는

버튼에넣는다. 그리고나서 2번프레임에다음코드를입력한다.

trace(x);

이렇게해도제대로작동할까? 물론이다. x가버튼에붙어있고버튼은메인타

임라인에 붙어 있기 때문에, 그 변수는 결국 메인 타임라인에 간접적으로 포함되는

셈이다. 따라서앞에서와마찬가지로 2번프레임에서변수를사용할수있다.

시나리오 3

메인 타임라인의 1번 프레임에서 secretPassword라는 변수를 만드는 경우를 생

각해보자. 무비를실행할때사용자가무비의특정부분을이용하려면암호를맞춰

야한다.

1번프레임에서 secretPassword를선언하는것외에사용자가입력한암호를실

제암호와비교하는함수도만든다. 코드는다음과같다.

var secretPassword;

secretPassword = "yppah";

function checkPassword() {

if (userPassword == secretPassword) {

gotoAndStop("accessGranted");

} else {

gotoAndStop("accessDenied");

}

}

30번 프레임에서 암호를 입력해야 한다고 가정하자. 사용자는 userPassword라

는 입력 텍스트 필드 변수에 암호를 입력한다. 그러면 1번 프레임에 있는 check

Password() 함수를 이용하여그값을 secretPassword 변수와비교한다. 암호 검사

코드는 1번 프레임에서 정의하지만 userPassword는 30번 프레임에 다다르기 전까

지는 정의되지 않는다. 그렇다면 우리가 checkPassword() 함수를 호출할 때는

userPassword 변수가존재할까?

이번에도 대답은‘그렇다’이다. userPassword를 checkPassword() 함수보다

나중에정의하더라도결국은그변수도같은타임라인에속하기때문이다.

변수 접근성(영역)

앞에서 다룬 세 개의 시나리오는 영역(scope)이라는 주제에 관한 것이다. 변수

의영역은무비에서언제, 그리고어디서그변수를조작할수있는지정해준다. 변

수의 영역은그변수의유효기간과스크립트에있는다른코드블록에서의접근성

을 정의한 것이다. 변수 영역을 결정하려면 두 가지 정보가 필요하다. (1) 변수가

얼마나오랫동안남아있을까? (2) 코드의 어느부분부터변수값을설정하거나검

색할수있을까?

전통적인 프로그래밍에서는 변수 영역의 범주를 크게‘전역(global) 변수’와

‘지역(local) 변수’로 나눈다. 프로그램 전체에서 사용할 수 있는 변수는 전역 변수

(global variable), 프로그램의제한된영역에서만사용할수있는변수는지역변수

(local variable)라고 부른다. 플래시에서는 일반적인 지역 변수는 지원하지만, 진정

한의미에서의전역변수는지원하지않는다. 그이유를알아보자.

무비 클립 변수

앞에서 본 세 가지 시나리오에서 알 수 있듯이 한 타임라인에서 정의된 변수는

같은 타임라인에 있는 모든 스크립트에서 사용할 수 있다. 프레임 맨 앞이나 맨 뒤

에있는코드라도상관없으며, 변수가프레임에서선언되었는지버튼에서선언되었

는지도 상관없다. 하지만 아래에 있는 시나리오 4와 같이 하나의 무비에 여러 개의

타임라인이있는경우에는어떻게될까?

정사각형과 원의 두 가지 기본 도형을 각각 무비 클립 심벌로 정의했다고 가정

하자.

직사각형클립심벌의 1번프레임에서 x라는변수값을 3으로설정한다.

var x;

x = 3;

원형클립심벌의 1번프레임에서 y라는변수값을 4로설정한다.

var y;

y = 4;

각 클립의 인스턴스를 무비메인 타임라인의 1번 레이어, 1번프레임에 갖다 놓

고각인스턴스의이름을 square와 circle이라고정한다.

첫째 질문: 다음과 같은 코드를 메인 무비 타임라인의 1번 프레임(square와

circle이있는곳)에덧붙인다면 Output 창에어떤내용이출력될까?

trace(x);

trace(y);

답: Output 창에 아무것도 나타나지 않는다. x와 y라는 변수는 무비 클립에 있

는타임라인에서정의된것일뿐, 메인타임라인에서정의된것이아니다.

둘째 질문: trace(x)와 trace(y)라는 선언문을메인무비타임라인의 1번프레임

이아닌 square 무비클립의 1번프레임에입력하면 Output 창에어떤내용이출력

될까?

답: x 값인 3만 출력된다. x는 square의 타임라인에서 정의되었기 때문에 같은

타임라인에 있는 trace() 명령어에서 사용할 수 있다. 하지만 y 값은 circle에서 정

의되어있으므로다른타임라인에속하기때문에사용할수없다.

이제액션스크립트에서전역변수를지원하지않는이유를이해할수있을것이

다. 전역 변수는 프로그램 전체에서 사용할 수 있어야 하는데, 플래시에서는 개별

타임라인에 들어 있는 변수는 같은 타임라인에 있는 스크립트에서만 직접 사용할

수있다. 플래시에있는모든변수는타임라인에서정의되므로무비에있는모든스

크립트에서직접사용할수있는변수는만들수없다. 따라서어떤변수도전역변

수라고부를수없다.

혼동을 피하기 위해 타임라인에 포함되는 변수를‘타임라인 변수(timeline

variable)’또는‘무비 클립 변수(movie clip variable)’라고 부르기로 하자. 하지만

Object 클래스를이용하면전역변수비슷한것을만들수있다. 모든타임라인에서

사용할수있는변수를만들려면다음과같은선언문을이용하면된다.

Object.prototype.myGlobalVariable = myValue;

예를들면다음과같다.

Object.prototype.msg = "Hello world";

이러한테크닉과원리에대한내용은‘12장. 객체와클래스’의‘상속사슬의끝’

절에서자세히다룬다.

다른 타임라인에 있는 변수 사용하기

어떤타임라인에들어있는변수를다른타임라인에있는스크립트에서직접사

용할수는없지만, 간접적으로사용하는것은가능하다. 다른타임라인에변수를만

들거나 사용하고 그 변수에 값을 대입할 때는 자바, C++, 자바스크립트 같은 객체

지향 프로그래밍 언어에서 흔히 쓰이는 표기법인 점 구문(dot syntax)을 사용하면

된다. 다른타임라인에있는변수를나타낼때사용하는일반적인점구문은다음과

같다.

movieClipInstanceName.variableName

즉그변수를포함하고있는클립의이름뒤에점을붙이고그뒤에변수이름을

붙이면, 다른 타임라인에있는변수를참조할수있다. 예를 들어앞에서살펴본시

나리오의경우를생각해보면메인타임라인에서 square 클립에있는변수 x를다음

과같이사용할수있다.

square.x

또한 circle 클립에있는변수 y는다음과같이참조할수있다.

circle.y

메인 무비 타임라인에서 위와 같은 레퍼런스를 사용하면 다음과 같은 방법으로

square에있는변수에새로운값을대입하거나저장되어있는값을알아낼수있다.

square.z = 5;       // sqaure에 있는 z에 5를 대입한다.

var mainZ;          // 메인 타임라인에 mainZ라는 변수를 만든다.

mainZ = square.z;   // mainZ에 square에 있는 z 값을 대입한다.

하지만 (클립 이름).(변수 이름)을 이용한 것만으로는 circle 클립에서 square 클

립에 있는 변수를 이용할 수 없다. circle에 있는 프레임에서 square.x를 사용하면

인터프리터에서는 circle 클립 안에 있는 square라는 클립을 찾으려고 하는데

square는메인타임라인에있는클립이기때문이다. 따라서 circle 클립에서 square

클립을 포함하고 있는 타임라인(이 경우에는 메인 타임라인)을 가리킬 수 있는 메커

니즘이 필요하다. 이러한 메커니즘은 두 개의 특별한 속성인 _root와 _parent를 이

용하여구현할수있다.

_root 속성은 무비의 메인 타임라인을 직접 참조하기 위한 레퍼런스이다. 무비

클립 구조에서어느위치에있더라도 _root를이용하면메인무비타임라인에있는

변수를참조할수있다.

_root.mainZ     // 메인 타임라인에 있는 mainZ라는 변수를 이용한다.

_root.firstName // 메인 타임라인에 있는 firstName이라는 변수를 이용한다.

_root를 이용하여 무비의 다단계 구조를 거쳐 임의의 무비 클립 인스턴스를 참

조하는 것도 가능하다. 예를 들어 메인 무비 타임라인에 있는 square 클립 안에 있

는변수 x는다음과같은식으로참조할수있다.

_root.square.x

이렇게 하면 무비 내의 어떤 클립에서든지 원하는 변수를 참조할 수 있다. 레퍼

런스가 메인 무비 타임라인인 _root에서부터 시작하기 때문이다. shapes라는 인스

턴스의 타임라인에 들어 있는 triangle이라는 인스턴스의 area라는 변수는 다음과

같이참조할수있다.

_root.shapes.triangle.area

_root 키워드로 시작하는 변수에 대한 레퍼런스는‘절대 레퍼런스(absolute

reference)’라고 부른다. 무비에서 고정되어 있고 바뀌지 않는 지점인 메인 타임라

인에서시작하여변수의위치를기술하는방식이기때문이다.

하지만무비의메인타임라인을참조하지않고다른타임라인에있는변수를참

조하고 싶을 때도 있다. 이럴 때는 현재 무비 클립 인스턴스를 포함하고 있는 타임

라인을 가리키는 _parent 속성을 이용하면 된다. 예를 들어 square 클립에 붙어 있

는 코드에서 square를 포함하고 있는 타임라인에 있는 변수를 참조할 때는 다음과

같이쓰면된다.

_parent.myVariable

_parent라는키워드로시작되는레퍼런스는‘상대레퍼런스(relative reference)’

라고 부른다. 현재 클립의 위치를 기준으로 상대적인 위치를 가리키는 방식이기 때

문이다.

앞에서사용한예제를다시생각해보자. 무비의메인타임라인에서정의한 size

라는변수가있다고가정하자. 메인 무비타임라인에 shapes라는이름의클립인스

턴스를 만들고 shapes 타임라인에 color라는 변수를 정의한다. 또 [그림 2-1]에 나

온것과같이 shapes 타임라인에 triangle이라는클립을만든다.

triangle 타임라인에있는코드에서(shapes 클립에있는) color 변수값을출력할

때는다음과같이메인타임라인에서시작하는절대레퍼런스를사용해도된다.

trace(_root.shapes.color);

하지만 이렇게 하면 코드가 메인 무비 타임라인에 얽매이게 된다. 코드를 더 유

연성있게만들고싶다면 _parent 속성을이용하여상대레퍼런스를만들면된다.

trace(_ parent.color);

_root를이용하는첫째방법에서는위에서아래로내려가는방식을사용하므로,

메인 타임라인에서 시작하여 무비 클립 계층구조를 따라 내려가서 color 변수를 찾

는다. _parent를사용하는둘째방법에서는밑에서위로올라가는방식을사용하므

로, trace() 선언문이들어있는클립(triangle 클립)에서시작하여클립구조를한단

계올라가서 color 변수를찾는다.

클립 계층에서 두 단계를 올라가서 메인 타임라인에 있는 size 변수를 이용하기

위해 _parent를두번연속으로사용해도된다. triangle에서다음과같은코드를사

용하면메인타임라인의 size를참조하게된다.

trace(_parent._parent.size);

_parent 속성을두번사용하면두단계를올라가므로이경우에는무비의메인

타임라인으로간다.

변수에 접근하는 방법은 무비 클립 심벌의 인스턴스를 다른 타임라인으로 옮겼

을 때 어떤 식으로 동작하기를 원하는지에 따라 달라질 수 있다. triangle 예제에서

만약 color에대한레퍼런스가언제나 shapes 클립에서정의된 color 변수를가리키

기를 원한다면, _root로 시작하는 구문을 사용하여 shapes에 있는 color에 대한 고

정된레퍼런스를이용하는것이낫다. 하지만 triangle이들어있는타임라인에따라

[그림 2-1] 무비 클립 계층구조의 예

메인 타임라인

size 변수

shapes 인스턴스

color 변수

triangle 인스턴스



상대적인 위치에 있는 color 변수를 참조하고 싶다면, _parent로 시작하는 구문을

이용하는것이낫다.

_root 속성은 현재단계(즉 현재 사용중인문서)의 메인 무비 타임라인을 가리키

지만, 플래시플레이어에서는문서스택(document stack)에 여러개의문서를한꺼

번에 담아둘 수 있다. 플레이어 문서 스택에 로딩된 임의의 무비 메인 타임라인은

_leveln(n은무비가위치하고있는레벨번호)을이용하여참조할수있다. 레벨번호

는 0에서시작하므로 _level0, _level1, _level2, _level3 같은식으로쓸수있다. 여

러 개의무비를로딩하는것과관련된내용은‘13장. 무비 클립’에나와있다. 다른

레벨에있는변수는다음과같은식으로참조하면된다.

_level1.firstName       // level1의 메인 타임라인에 있는 firstName

_level4.ball.area       // level4의 메인 타임라인에 있는 ball 클립의 area

_level0.guestBook.email // level10의 타임라인에 있는 guestBook 클립의 email

/square:area와같이플래시 4 스타일의슬래시와콜론을이용하는구조는플래

시 5에서는 점 구문으로 바뀌었다. 이 방법이 변수와 타임라인을 참조하기 훨씬 좋

기 때문이다. 예전에 쓰던 문법은 이제는 더 이상 쓰이지 않기 때문에 새로 만드는

프로그램에서는 쓰지 않는 것이 좋다. [표 2-1]에 플래시 4와 플래시 5에서 변수를

참조하는 방법을 비교해 놓았다. 문법적인 차이점에 대한 자세한 내용은‘부록 C.

하위호환성’을참조하기바란다.

점 구문을 이용하여 여러 무비 클립 인스턴스 타임라인에 걸쳐 변수를 이용할 때는 스테이
지에서 클립 인스턴스에 이름을 제대로 붙였는지, 코드에서 변수 이름을 정확하게 입력했
는지 확인하도록 하자. 인스턴스에 이름이 없으면 코드 자체가 문법적으로 완벽하다고 하
더라도 제대로 동작하지 않게 된다. 인스턴스의 이름이 없거나 이름을 잘못 적는 일로 인
해 자주 문제가 발생한다.



[표 2-1] 플래시 4와 플래시 5의 변수 사용 문법

플래시 4 문법 플래시 5 문법 참조하는 내용

/ _root 무비의 메인 타임라인

/:x _root.x 무비의 메인 타임라인에 있는 변수 x

/clip1:x _root.clip1.x 무비의 메인 타임라인에 있는 clip1의 변수 x

/clip1/clip2:x _root.clip1.clip2.x 메인 무비 타임라인의 인스턴스 clip1에 포함된 인스턴
스 clip2에 있는 변수 x

../ _parent 현재 클립 바로 위에 있는 타임라인(현재 클립 타임라
인*의 한 단계 위)

../:x _parent.x 현재 클립 바로 위에 있는 타임라인에 있는 변수 x

../../:x _parent._parent.x 현재 클립을 포함하는 타임라인을 포함하는 타임라인
(현재 클립 타임라인에서 두 단계 위)에 있는 변수 x

clip1:x clip1.x 현재 타임라인에 들어 있는 인스턴스인 clip1에 들어 있
는 변수 x

clip1/clip2:x clip1.clip2.x 현재 타임라인에 포함된 clip1에 포함된 clip2에 들어 있
는 변수 x

_level1:x _level1.x 레벨 1에 로딩된 무비의 메인 타임라인에 있는 변수 x

_level2:x _level2.x 레벨 2에 로딩된 무비의 메인 타임라인에 있는 변수 x

* 현재 클립 타임라인은 변수 레퍼런스를 포함하고 있는 코드가 포함된 타임라인을 뜻한다.

무비 클립 변수의 생존 기간

앞에서변수영역에의해두가지질문에답할수있다는것을배웠다. (1) 변수

가 얼마나오랫동안남아있을까? (2) 코드의 어느부분부터변수값을설정하거나

검색할수있을까? 무비클립변수의경우둘째질문에대한답은이제알고있으리

라생각한다. 하지만첫째질문의답은아직잘모른다. 마지막변수코딩시나리오

를통해그문제를해결해보자.

두 개의 키프레임이 있는 새로운 무비를 만든다고 가정하자. 1번 프레임에 ball

이라는클립인스턴스를추가한다. ball 타임라인에서 radius라는변수를만든다. 메

인타임라인의 2번프레임은비어있는채로둔다(여기에는 ball 인스턴스가없다).

메인 무비 타임라인의 1번 프레임에서는 다음과 같은 코드를 이용하여 radius

값을알아낼수있다.

trace(ball.radius);

질문: 만약 위와 같은 코드를 메인 타임라인의 1번 프레임에서 2번 프레임으로

옮겨놓는다면무비를재생할때 Output 창에어떤내용이출력될까?

답: 아무것도보이지않는다. ball 클립을 2번프레임의메인타임라인에서없애

면그과정에서모든변수가제거된다.

다양한타임라인에서여러개의프레임에걸쳐있는무비클립을포함하는무비

스크립트를만들때는변수의생존기간이매우중요하다. 어떤클립에있는변수를

사용하기전에사용하려는클립이타임라인에존재하는지반드시확인하도록하자.

지역 변수

무비 클립 변수는 무비 클립 영역 안에서 사용할 수 있으며, 그 변수가 속해 있

는무비클립을사용중이라면언제든지사용할수있다. 경우에따라필요이상으로

변수가 오래 남아 있는 경우도 있다. 임시로 쓸 변수가 필요한 경우를 위해 액션스

크립트에서는 지역 변수를 제공하는데, 지역 변수는 일반적인 무비 클립 변수에 비

해짧은기간동안만남아있는다.

지역 변수는함수또는예전에쓰이던플래시 4 스타일의서브루틴에서쓰인다.

함수나 서브루틴을 사용해 본 적이 없다면, 이 절의 나머지 부분은 일단 그냥 넘어

가고‘9장. 함수’를마친후에다시읽어보는것이낫다.

함수에서는 종종 함수 밖에서는 필요 없는 변수를 사용하는 경우가 있다. 예를

들어주어진배열의모든원소를화면에표시하는함수를생각해보자.

무비 클립 변수는 그 변수가 들어 있는 클립이 스테이지에 있는 동안만 존재할 수 있다. 플래
시 문서의 메인 타임라인에서 정의한 변수는 각 문서에서 계속해서 사용할 수 있지만, 플레이
어에서 문서를 닫고 나면 모두 사라진다(unloadMovie( ) 함수를 이용하거나 같은 레벨에 다른
무비를 로딩하는 경우에 상관없이 같은 결과가 나타난다).



function displayElements(theArray) {

var counter = 0;

while(counter < theArray.length) {

trace("Element " + counter + ": " + theArray[counter]);

counter++;

}

}

counter라는변수는배열을표시하기위해필요하지만그작업이끝나고나면쓸

모가 없어진다. 이 변수를 타임라인에 그냥 남겨 둬도 되지만 다음과 같은 두 가지

이유에서그렇게하지않는것이좋다. 첫째, counter 변수가계속남아있으면무비

가끝날때까지계속메모리를차지하게된다. 둘째, counter를함수밖에서도사용

할 수 있다면 같은 이름을 가진 다른 변수와 충돌할 수도 있다. 따라서

dispalyElements() 함수가끝나고나면 counter 변수가없어지도록하는것이좋다.

함수가 끝날때 counter 변수도자동으로없어지게하려면그변수를지역변수

로정의하면된다. 무비 클립변수와달리지역변수는그변수를정의한함수가완

료되면인터프리터에의해자동으로제거된다.

어떤변수를지역변수로지정하려면앞에나온 displayElements() 예제에서한

것처럼함수내에서 var 키워드를이용하여선언하면된다.

이 때 주의할 점이 있다. 만약 var 선언문을 함수 밖에서 사용한다면 지역 변수

가 아닌 일반적인 타임라인 변수가 만들어진다. [예제 2-4]에서 볼 수 있듯이 var

선언문의위치에따라결과가크게달라진다.

함수안에있는변수가반드시지역변수일필요는없다. var 키워드를생략하면

함수안에서도무비클립변수를만들거나그값을변경할수있다. 변수를만들때

var 키워드를사용하지않고그냥변수에어떤값을대입하면플래시에서는상황에

따라그변수를지역변수가아닌것으로간주한다. 아래와같이함수안에있는변

수대입선언문을생각해보자.

function setHeight(){

height = 10;

}

height = 10;이라는 선언문에 의한 결과는 height가 지역 변수인지, 아니면 무

비클립변수인지에따라달라진다. 만약 height가전에선언된지역변수라면(물론

위에 있는예제의 경우에는그렇지않다), height = 10;이라는 선언문에서는 단순히

지역 변수의 값을 변경하게 된다. 만약 위 예제처럼 height라는 이름을 가진 지역

변수가없다면인터프리터에서는 height라는이름의무비클립변수(지역변수가아

닌 변수)를 만들고 그 값을 10으로 설정한다. 지역 변수가 아닌 경우에는 height라

는변수는함수가끝나고난뒤에도그대로남아있게된다.

[예제 2-4]에지역변수와지역변수가아닌변수의사용법을정리해보았다.

[예제 2-4] 지역 변수와 지역 변수가 아닌 변수

var x = 5;             // 새로운 지역 변수가 아닌 변수. x의 값은 5

이다.

function variableDemo(){

x = 10;           // 지역 변수가 아닌 변수 x. 값이 10이 된다.

y = 20;           // 새로운 지역 변수가 아닌 변수 y. 값은 20

이다.

var z = 30;       // 새로운 지역 변수 z. 값은 30이다.

trace(x + "," + y + "," + z); // 모든 값들을 Output 창에 출력한다.

}

variableDemo();// 위의 함수를 호출하면 10,20,30 이 출력된다.

trace(x);  // 10이 출력됨(함수 내에서 새로운 값을 대입하면 그 값이 유지된다)

trace(y);     // 20이 출력됨(y는 지역 변수가 아니므로 그대로 남아 있다)

trace(z);   // 아무 것도 출력되지 않는다(지역 변수인 z는 이미 사라졌다).

조금 혼동되기도 하고 그다지 좋지 않은 방법이긴 하지만, 같은 스크립트에서

이름은같지만영역이다른지역변수와지역변수가아닌변수를동시에사용하는

것도가능하다. [예제 2-5]에서그러한경우를볼수있다.

var myColor = "blue";

function hexRed(){

var myColor = "#FF0000";

return myColor;

}

trace(hexRed()); // #FF0000이 출력된다(지역 변수 myColor).

trace(myColor);  // "blue"가 출력된다(지역 변수인 myColor를 #FF0000으로
// 설정하더라도 지역 변수가 아닌 변수는 영향을 받지 않는다).

이식성이좋은코드모듈을만들때는함수를사용하는것이좋지만, 플래시 5에

서도 플래시 4 스타일의 서브루틴(subroutine)을 지원하기는 한다. 플래시 4에서는

레이블이있는프레임에코드블록을덧붙여서서브루틴을만들수있다. 그렇게만

들어두면 Call 액션을통해서브루틴을외부에서실행시킬수있다. 하지만플래시

4에서는 서브루틴에서 선언한 모든 변수가 지역 변수가 아니기 때문에, 그 변수를

정의한 타임라인이 남아 있는 동안 그 변수들이 모두 그대로 남아 있게 된다. 플래

시 5에서는 함수에서 사용한 것과 같은 방법으로 var 선언문을 이용하여 서브루틴

에서도 지역 변수를 만들 수 있다. 하지만 서브루틴에서 var를 이용하여 정의한 변

수는서브루틴을 Call 함수를이용하여실행한경우에만지역변수가된다. 만약 서

브루틴프레임의스크립트가플레이헤드가그프레임에들어갈때자동으로실행되

는경우에는 var 선언문을이용하더라도일반적인타임라인변수가된다. 하지만이

런기능이지원된다고하더라도서브루틴보다는더최근에나온방법인함수및함

수에포함된지역변수를사용하는것이좋다.

몇 가지 응용 예제
지금까지 변수에 대한 이론적인 내용을 꽤 많이 배웠다. 이러한 개념을 실전에

응용해 보자. 아래에 세 개의 변수와 관련된 예제를 수록하였다. 코드에 대한 자세

한설명은코드주석을참조하기바란다.

[예제 2-6]에서는무비의플레이헤드를임의의위치로옮긴다.

[예제 2-6] 무비의 플레이헤드를 현재 타임라인에 있는 임의의 위치로 옮긴다.

var randomFrame;            // 임의로 만들어낸 프레임 번호를 저장하는 변수
var numFrames;              // 타임라인의 전체 프레임 번호를 저장하는 변수
numFrames = _totalframes;   // numFrames에 _totalframes 속성을 저장한다.

// 임의의 프레임을 선택한다.

randomFrame = Math.floor(Math.random() * numFrames + 1);

gotoAndStop(randomFrame);   // 플레이헤드를 선택한 프레임으로 보낸다.

[예제 2-7]에서는두클립사이의거리를알아낸다. 이 예제를실제로응용한프

로그램은온라인코드창고에서구할수있다.

[예제 2-7] 두 무비 클립 사이의 거리를 계산한다.

var c;         // circle 클립 객체에 대한 레퍼런스
var s;         // square 클립 객체에 대한 레퍼런스
var deltaX;    // c와 s의 수평방향 거리
var deltaY;    // c와 s의 수직방향 거리
var dist;      // c와 s 사이의 전체 거리

c = _root.circle;         // circle 클립에 대한 레퍼런스를 구한다.

s = _root.square;         // square 클립에 대한 레퍼런스를 구한다.

deltaX = c._x - s._x;     // 클립 사이의 수평 거리를 구한다.

deltaY = c._ y - s._ y;   // 클립 사이의 수직 거리를 구한다.

// 거리 = (deltaX의 제곱 더하기 deltaY의 제곱).

dist = Math.sqrt((deltaX * deltaX) + (deltaY * deltaY));

// 깔끔하게 레퍼런스를 이용하면 코드를 읽기가 훨씬 편하다.

dist = Math.sqrt(((_root.circle._x - _root.square._x) * (_root.circle._x -

_root.square._x)) + ((_root.circle._ y - _root.square._ y) *

(_root.circle._ y -

_root.square._ y)));

[예제 2-8]에서는 화씨 온도와 섭씨 온도를 변환한다. 온라인 코드 창고에서 실

행해볼수있는버전을구할수있다.

[예제 2-8] 화씨/섭씨 온도 변환

var fahrenheit;         // 화씨 온도
var celsius;            // 섭씨 온도
var convertDirection;   // 변환하고자 하는 온도 표기법

// "fahrenheit(화씨)", "celsius(섭씨)" 중 한 값을
// 가질 수 있다.

fahrenheit = 451;                // 화씨 온도를 설정한다.

celsius = 20;                    // 섭씨 온도를 설정한다.

convertDirection = "celsius";    // 이 경우에는 섭씨로 변환

if (convertDirection == "fahrenheit") {

result = (celsius * 1.8) + 32;   // 섭씨 온도 계산

// 결과 출력
trace (celsius + "degrees Celsius is" + result + "degrees Fahrenheit.");

} else if (convertDirection == "celsius") {

result = (fahrenheit - 32) / 1.8;   // 화씨 온도 계산
// 결과 출력
trace (fahrenheit+"degrees Fahrenheit is"+result+"degrees Celsius.");

} else {

trace ("Invalid conversion direction.");

}

앞으로 배울 내용
이제변수에정보를저장하는것은모두배웠으므로변수에저장할내용물인데

이터에 대해 배워 보자. 앞으로 나올 세 장에서는 데이터가 무엇인지, 데이터를 어

떻게 조작하는지, 그리고 그런 작업이 왜 액션스크립트에서 필수적인 부분이 되는

지배울것이다.

